

Paper 201-30

## Benchmark Dose Estimation Using SAS®

**Matthew W. Wheeler, National Institute for Occupational Safety and Health**

### ABSTRACT

Toxicologically-based quantitative risk assessment is concerned with estimating human risks based upon experimental data linking an environmental agent to a known outcome (tumor incidence, acute toxicity, etc.). For dichotomous outcomes dose-response curves are modeled as complex functions of dose which often require specialized software to estimate.

The SAS procedure NLMIXED readily allows for maximum likelihood estimation of binomial response data to any non-linear function. This power gives the SAS system the ability to fit dichotomous response curves that have traditionally been modeled using specialized software. For this paper excess risk is estimated within SAS using NLMIXED and the SAS MACRO programming language. Nine dose-response curves are fit examining the excess risk associated with renal tubular degeneration in ethylene glycol exposed rats. The results are compared with those given using the USEPA Benchmark Dose Software. The results show that these models can be reliably implemented within the SAS system producing very similar results as the USEPA software.

### INTRODUCTION

Toxicologically based quantitative risk assessment is concerned with estimating human risks based upon experimental data linking an environmental hazard to a known outcome (tumor incidence, acute toxicity, etc.). Risk, the probability of some adverse response, is often derived from dose-response models which parameterize risk as a function of dose. These models are commonly fit using maximum likelihood (ML) estimation. The complexities of the likelihood equation and corresponding risk estimation have forced investigators to use specialized software outside of SAS to conduct their analyses. This is not only an inconvenience for experienced SAS users, but prevents the modeler from incorporating techniques which are not directly implemented by the third party software. For example, if one wanted to examine the behavior of a given dose-response curve through Monte Carlo simulation this task is overly cumbersome, given current software, as most risk estimation packages do not provide the ability to generate random data sets.

This paper examines SAS as a software package which can be used to model risk. Specifically we demonstrate the ability of NLMIXED to model risk derived from dichotomous data. Doses associated with a specified level of excess risk above control response, so-called benchmark doses, are estimated as well. This is done using the procedure NLMIXED with help from the MACRO programming language. Nine popular models, which are also parameterized in the United

States Environmental Protection Agency (USEPA) Benchmark Dose Software [1], are fit and compared to results from that software package.

## MODELING RISK

Toxicologically based risk assessment frequently models the probability of adverse outcome,  $\pi$ , as a function of dose,  $d$ . Though this dose-response relationship can be modeled through a wide class of functions, we restrict our discussion to implementing nine popular dose-response curves, which are also fit by the USEPA Benchmark Dose Software [1]. These models are as follows

$$\text{logistic:} \quad \pi(d) = \frac{1}{1 + \exp[-(\alpha + \beta d)]} \quad (1)$$

$$\text{log-logistic :} \quad \pi(d) = \gamma + \frac{(1 - \gamma)}{1 + \exp[-(\alpha + \beta \ln(d))]} \quad (2)$$

$$\text{gamma:} \quad \pi(d) = \gamma + (1 - \gamma) \frac{1}{\Gamma(\alpha)} \int_0^{\beta d} t^{\alpha-1} e^{-t} dt \quad (3)$$

$$\text{multistage} \quad \pi(d) = \gamma + (1 - \gamma) [1 - \exp(-f(\underline{\theta}, d, n))] \quad (4)$$

$$\text{probit} \quad \pi(d) = \Phi(\alpha + \beta d) \quad (5)$$

$$\text{log-probit} \quad \pi(d) = \gamma + (1 - \gamma) \Phi[\alpha + \beta \ln d] \quad (6)$$

$$\text{quantal-linear} \quad \pi(d) = \gamma + (1 - \gamma) [1 - \exp(-\beta d)] \quad (7)$$

$$\text{quantal-quadratic} \quad \pi(d) = \gamma + (1 - \gamma) [1 - \exp(-\beta d^2)] \quad (8)$$

$$\text{weibull} \quad \pi(d) = \gamma + (1 - \gamma) [1 - \exp(-\beta d^\alpha)] \quad (9)$$

where  $\Gamma(\alpha)$  = gamma function evaluated at  $\alpha$ ,  $\Phi(x)$  = CDF for  $N(0,1)$  and  $\pi_i = \gamma$  when  $d=0$  for models (2) and (6). Further  $f(\underline{\theta}, d, n)$  in model (4) is an  $n$  degree polynomial of the dose  $d$  having the vector of coefficients  $\underline{\theta}$ . Models (1)-(9) are fit using maximum likelihood estimation. Further, in models (2),(4) and (6)-(8) the slope terms  $(\beta, \underline{\theta})$  are bounded to be non-negative, in models (3) and (9) the power term  $(\alpha)$  is bounded below by one, and the background response term  $(\gamma)$  for models (2)-(4) and (6)-(9) is bounded between zero and one.

For any particular dose-response function excess risk is often characterized through the use of the benchmark dose [2]. A Benchmark dose (BMD) is defined as the dose that increases risk over the background response rate by some amount relative to the control response. For dichotomous outcomes this pre-specified level, known as the benchmark response (BMR), is commonly given values of 1 and 10%. Given this value of the BMR and a model representing the dose-response [e.g. models (1)-(9)] excess risk can be found by finding the dose ( $BMD$ ) which satisfies the following equation

$$BMR = \frac{\pi(BMD) - \pi(0)}{1 - \pi(0)} \quad (10)$$

where  $\pi(0)$  is the background response. The above formulation is known as the extra risk, and can be thought of as the probability that an effect is observed among individuals who would not have the adverse response in the absence of exposure to the environmental hazard. Though other definitions of excess risk exist, specifically added risk, and are implemented in the USEPA software, for illustration purposes the extra risk formulation is used in the following discussion. It should be noted that the added risk formulation can also be programmed into NLMIXED as it is very similar to the extra risk formulation.

### RENAL TUBULAR DEGENERATION DATA ANALYSIS

Consider a 10 day exposure study where Sprague-Dawley rats were exposed to ethylene glycol (EG) in their drinking water [3], with the number of rats exhibiting renal tubular degeneration measured as the observed response. In this experiment, ten rats were exposed to one of 5 dose groups (0, 0.5, 1.0, 2.0 and 4.0% EG). The observed proportions of rats exhibiting renal tubular degeneration were 2/10, 2/10, 2/10, 6/10, and 9/10 for EG concentrations of 0, 0.5, 1.0, 2.0, and 4.0%. The data are analyzed using models (1-9).

For illustration, consider the Weibull model (9) and the definition for extra risk given in equation (10). The SAS procedure NLMIXED can provide ML estimates for this model's parameters as well as estimates for the BMD. The parameters are estimated directly through the use of the MODEL statement, but because the BMD is specified through model parameters, its ML estimate must be computed through programming statements which algebraically relate the ML estimates for the model parameters to the BMD. In the case of the assumed model NLMIXED programming statements estimate the BMD using the following equation

$$\hat{BMD} = \left[ \frac{-\ln(1 - BMR)}{\hat{\beta}} \right]^{1/\hat{\alpha}} \quad (11)$$

which is found by substituting the Weibull model (9) into the extra risk equation (10) and algebraically solving for the  $BMD$  term. The following NLMIXED code which implements the above, subject to the bounds on the parameters of  $\alpha \geq 1, \beta \geq 0$  and  $0 \leq \gamma \leq 1$ .

```

001  data EGdata;
002      input dose obs n;
003      cards;
004  0 2 10
005  0.5 2 10
006  1.0 2 10
007  2.0 6 10
008  4.0 9 10
009  ;
010
011  %LET BMR = 0.1;
012
013  PROC NLMIXED data= EGdata;
014      PARSMS _GAMMA = 0.006 _BDOSE = 0.5 _ALPHA = 1;
015      BOUNDS _GAMMA >= 0, _GAMMA <= 1, _ALPHA >= 1, _BDOSE >= 0;
016      P = _GAMMA;
017      IF (DOSE > 0) THEN DO;
018          _LINK = _BDOSE * (DOSE**_ALPHA);
019          P = _GAMMA + (1-_GAMMA)*(1-EXP(-_LINK));
020      END;
021      _X = -LOG(1-&BMR);
022      _BMD = (_X/_BDOSE)**(1/_ALPHA);
023      CALL SYMPUT("BMD",_BMD);
024      MODEL OBS ~ BINOMIAL(N,P);
025  RUN;

```

Though NLMIXED can be used to model non-linear mixed models and has a variety of features to aid in this facility, the above code only utilizes its maximum likelihood estimation ability, which include the statements PARSMS, BOUNDS, and MODEL. These programming statements, which are described in the documentation [4], set up the initial parameter estimates, the bounds on the parameters, and the likelihood equation respectively. The Weibull dose-response curve in the above code is estimated through programming statements found in lines 16-20, where the variable “dose” represents the amount of exposed EG, “obs” represents the number observed rats exhibiting renal tubular degeneration, and n is the number of rats in each dose group. Further lines 21-23 algebraically specify the ML estimate for the benchmark dose, and line 24 specifies the likelihood equation. Though it is possible to output the BMD estimate to a dataset using the PREDICT statement, this is not done in favor of outputting the final value to a MACRO variable labeled BMD, which is referenced as &BMD; this variable is created in line 25.

Though the above code is sufficient to allow NLMIXED’s algorithm to converge using the EG data some data sets may require more precise starting values. In this case, a grid search is often sufficient for most problems which require more precise starting values. The above code produces the following estimates which correspond within three significant figures to the output produced by the USEPA Benchmark Dose Software.

TABLE 1.

|                | NLMIXED |                | USEPA  |                |
|----------------|---------|----------------|--------|----------------|
|                | MLE     | Standard Error | MLE    | Standard Error |
| $\hat{\gamma}$ | 0.1724  | 0.08877        | 0.1724 | 0.08877        |
| $\hat{\alpha}$ | 2.0275  | 0.8264         | 2.0275 | 0.8263         |
| $\hat{\beta}$  | 0.1373  | 0.1382         | 0.1373 | 0.1382         |

### CONFIDENCE INTERVALS FOR THE BENCHMARK DOSE

Though NLMIXED can be used to estimate the BMD as well as asymptotic confidence intervals, using the PREDICT statement, it is not advisable to use the confidence intervals provided by this facility of NLMIXED. These intervals are based upon standard large sample asymptotic approximations, which may not be appropriate in small sample situations. Instead Crump and Howe [5] suggest the use of the profile likelihood in computation of the confidence interval for the benchmark dose, as it generally gives better coverage behavior in such conditions.

It is convenient to look at the profile likelihood in terms of a likelihood ratio test, and remember the fact that the  $-2 \log$  of the likelihood ratio follows an approximate chi-squared distribution with  $n$  degrees of freedom, where  $n$  represents the number of parameters fixed under the null hypothesis. A profile likelihood confidence interval, in essence, is a likelihood ratio test where one finds a “null” model that fixes the parameter of interest while allowing the other model parameters to vary such that the  $-2 \log$ -likelihood of this model is different from the  $-2 \log$ -likelihood of the original or “full” model by  $\chi^2_{(1,1-\alpha)}$ . Where  $\chi^2_{(1,1-\alpha)}$  represents the  $(1-\alpha)^{\text{th}}$  quantile of a chi-squared distribution with one degree of freedom. Thus the two distinct values that fix the parameter of interest and reduce the maximum likelihood by  $\chi^2_{(1,1-\alpha)}$  represent the  $100(1-\alpha)\%$  upper and lower confidence bounds on this parameter. In our context the lower bound of the benchmark dose (BMDL) is found by searching for the BMD that changes the  $-2 \log$ -likelihood of the full model by  $\chi^2_{(1,1-2\alpha)}$ , where the  $1-2\alpha$  is chosen because the confidence interval of interest is one sided. Though models (1)-(9) do not explicitly parameterize the BMD, they can be parameterized as a function of the BMD.

It is not possible to directly estimate profile likelihood confidence intervals using NLMIXED; however one can use a combination of the MACRO programming language and the Output Delivery System to estimate the proper  $100(1-\alpha)\%$  profile likelihood based confidence interval. This is achieved by fixing a parameter in the model relative to the BMD and maximizing the likelihood relative to the other free parameters in the model. Again for illustration we examine the Weibull model. In this situation we set  $\beta$  equal to a function of the BMD i.e.

$$\beta = \frac{-\ln(1 - BMR)}{BMD^\alpha}. \quad (12)$$

By using equation (12) one can decrease the value of the BMD until the resultant -2 log-likelihood is greater than  $\chi^2_{(1,1-2\alpha)}$ . This is done using the macro programming language as follows.

```

001 %macro boundBMD(BMDL,MLIKE,CL);
002 %let BMR = 0.1;
003 %let MLIKE = %sysevalf(&mlike/2);
004 data _temp_;
005         val = cinv(1-2*(1-&CL),1)*0.5;
006         call symput("CRITVAL",val);
008 run;
009 %LET CRITLIKE = &MLIKE;
010
011 %DO %WHILE (%SYSEVALF(&CRITLIKE - &MLIKE < &CRITVAL));
012     *set up the initial parameters for the new likelihood;
013     %LET FIRSTTOKEN = 1;
014     %LET LASTBMDL = &BMDL;
015     %LET BMDL = %SYSEVALF(0.98*&BMDL);
016     %LET BOUNDS = _GAMMA >= 0, _GAMMA <= 1, _ALPHA >= 1, _ALPHA <= 18;
017     %LET SMODEL = ;
018 data pe; set ParameterEstimates; if(Parameter='_BDOSE')then delete;
019 run;
020     data pe;
021         set pe;
022     run;
023     ods listing close;
024     ods output fitstatistics = fitstatistics
025             ParameterEstimates=ParameterEstimates;
026     *fit this new "constrained" likelihood;
027     proc nlmixed data= one;
028         parms /data=pe; ;
029         bounds &bounds;
030
031         _X = (-LOG(1-&BMR))*(1/_ALPHA);
032         *solve for the BETA parameter BDOSE as a function of the BMD;
033         _BDOSE = (_X/&BMDL)**_ALPHA;
034         P = _GAMMA;
035         IF (DOSE > 0) THEN DO;
036             _LINK = _BDOSE*DOSE**_ALPHA;
037             P = _GAMMA + (1-_GAMMA)*(1-EXP(-_LINK));
038         END;
039         MODEL OBS ~ BINOMIAL(N,P);
040 run;
041     ods listing;
042     *obtain the Fit statistics to determine if the algorithm has bounded
043     the BMDL;
044     data fitstatistics;
045         set fitstatistics;
046         format value best16.;
047         informat value best16.;
048         if (Descr = "-2 Log Likelihood");
049         NegLogLike = value/2;
050         keep value NegLogLike;
051 run;
052     data _temp_;
053         set fitstatistics;
054         call symput("CRITLIKE",NegLogLike);
055 run;
056 %END;
057 %put &bmdl;
058 %mend;

```

To invoke this macro one needs to have the ML estimate of the BMD, the -2 log-likelihood of the unconstrained ML and the desired 100(1- $\alpha$ )% confidence level. These values are represented by the macro variables BMDL, MLIKE, and CL respectively. Further, to link it with the code used for the EG data one would need to specify the parameter estimates of the maximum likelihood in a data set named ParameterEstimates; an example of retrieving this dataset is found within the macro on line 24.

The above macro iteratively lowers the estimate of the BMDL by 2% until -2 log-likelihood is reduced by  $\chi^2_{(1,1-2\alpha)}$  as stored in &CRITVAL. Lines 13-22 initialize the variables for the current iteration, 27-40 use NLMIXED to optimize the likelihood with the “ $\beta$ ” constrained by the BMD, and lines 42-55 retrieve the necessary statistics from the data sets to determine the stopping criteria. Though this procedure puts a lower bound on the BMD it does not produce a valid 100(1- $\alpha$ )% confidence level. To find the approximate 100(1- $\alpha$ )% confidence interval one can take the value from the lower bound procedure described above, the ML estimate, along with a root finding algorithm to find this value to any specified level of tolerance. In the context of the profile likelihood one finds the root of the function

$$-2\ln(l(\hat{BMD}_{profile})) + 2\ln(l(\hat{BMD}_{ml})) - \chi^2_{(1,1-\alpha)} \quad (13)$$

where  $l(\hat{BMD}_{ml})$  represents the likelihood function evaluated at the ML and  $l(\hat{BMD}_{profile})$  represents the likelihood function evaluated at the current estimate of the BMD lower bound.

The above code has been packaged with a root finding procedure in one macro which estimates dose-response model parameters, the BMD, and BMDL given the BMR with a specified confidence level. These macros have been applied to the EG data using the 9 dose-response and produce the following output.

**TABLE 2.**

| <b>Model</b>             | <b>BMD<br/>10%</b> | <b>BMDL<br/>10%</b> | <b>BMD<br/>1%</b> | <b>BMDL<br/>1%</b> |
|--------------------------|--------------------|---------------------|-------------------|--------------------|
| Logistic                 | 0.59               | 0.40                | 0.07              | 0.04               |
| Log-Logistic             | 1.07               | 0.39                | 0.52              | 0.08               |
| Gamma                    | 1.01               | 0.28                | 0.47              | 0.03               |
| Multi-Stage<br>Quadratic | 0.86               | 0.25                | 0.27              | 0.02               |
| Probit                   | 0.56               | 0.40                | 0.06              | 0.04               |
| Log-Probit               | 1.10               | 0.42                | 0.64              | 0.15               |
| Quantal-<br>Linear       | 0.29               | 0.18                | 0.03              | 0.02               |
| Quantal-<br>Quadratic    | 0.86               | 0.65                | 0.27              | 0.20               |
| Weibull                  | 0.88               | 0.26                | 0.28              | 0.03               |

The estimates of table 2 correspond within three significant figures to the estimates from the USEPA Benchmark Dose Software.

## ESTIMATION ABILITY

Consider a hypothetical experiment where four groups of 50 rats are exposed to an environmental agent and mortality, which is recorded at the end of the two year study, is the measured response. In this experiment, the four dose groups represent 0, 25, 50 and 100% of the maximum dose administered, with mortality proportions of 4/50, 11/50, 32/50, and 50/50 corresponding respectively to the dose-groups (e.g. 0, 25, 50 and 100%). If one were to fit the two-stage multistage model (4) using the USEPA Benchmark Dose Software, the software would fail to converge to the ML estimate. However if one were to model risk using the procedures outlined above one would obtain the ML estimates for all parameters as well as the benchmark dose value 0.16 with the 95% lower bound of this estimate being 0.12.

## CONCLUSION

The SAS procedure NLMIXED allows for the fitting of many common BMD models. Though NLMIXED alone is not sufficient to find proper confidence intervals the MACRO programming language in combination with NLMIXED can be used effectively to compute profile likelihood based confidence intervals. Further the NLMIXED procedure has been shown to find the maximum likelihood when other models fail to find the optimum. Finally, as SAS is a multipurpose statistical package, it is versatile enough to extend present BMD software capabilities beyond those provided by specialized software.

## ACKNOWLEDGEMENTS

The author would like to thank Jim Bena, Steve Gilbert, Drs. John Bailer, Gregg Dinse, and Bob Nobel for their comments on an early draft of this paper.

## REFERENCES

1. USEPA. 2001. Help Manual for Benchmark Dose Software, Version 1.3. USEPA, Research Triangle Park, NC, EPA 600/R-00/014F.
2. Crump K. 1984. A new method for Determining Allowable Daily Intakes *Fundamental And Applied Toxicology* 4: 854-871
3. Robinson M., C.L. Pond, R.D. Laurie, J.P. Bercz, G. Henningsen, and L.W. Condie. 1990. Subacute and subchronic toxicity of ethylene glycol administered in drinking water to Sprague-Dawley rats. *Drug Chem. Toxicol.* 13(1):43-70.
4. SAS Institute Inc.1999. *SAS/STAT User's Guide, Version 8*. Cary, NC: SAS Institute Inc.

5. Crump, K.S. and R. Howe. 1985. A review of methods for calculating statistical confidence limits in low dose extrapolation in Clayson, D. Krewski, I. Munro(eds). *Toxicological Risk Assessment*, Vol. I (pp.187-203). Boca Raton, Fl: CRC Press.

### **CONTACT INFORMATION**

For information about this paper or SAS code which fits models (1)-(9) please contact:

Matthew W. Wheeler  
Risk Evaluation Branch  
National Institute for Occupational Safety and Health  
4676 Columbia Parkway  
Cincinnati, OH 45224  
(513) 533-8195  
MWheeler@cdc.gov

SAS and all other SAS Institute Inc. product or service names are registered trademarks or trademarks of SAS Institute Inc. in the USA and other countries. ® indicates USA registration.

Other brand and product names are trademarks of their respective companies.